


Abstract

Column A de planilha contém os valores numéricos, que devem ser ordenados em outra coluna, usando somente as funções de Excel. Nós propomos essa ordenação usando somente as funções: SE, INDIRETO, CONCATENAR, LINHA, ERROS, CORRESP.

1 Descrição de problema

Suponha a coluna A tem números, que comecem a A2 (deixamos a primeira linha para comentários etc.). Preenchemos as células de A com formulas



	A	B	C	D
1		bloco		
2	378	1		
3	775	1		
4	372	2		
5	90	3		
6	387	3		
7	211	4		
8	86	5		
9	175	5		
10	68	6		
11	521	6		
12	635	6		
13	553	7		
14	379	8		

Figure 1:

=ALEATÓRIOENTRE(0; 1000)

para os testes futuros. Vamos adicionar a coluna B com as formulas seguintes:

B1: 1

B2: =SE(A3>=A2;B2;B2+1)

Copiamos B2 para baixo.

Assim, escrevemos na coluna B os números de blocos ordenados de coluna A. Se coluna A é ordenada, crescente, todos os valores de B serão 1s. No caso contrário A precisa de uma permutação para ser ordenada. Mais disso, podemos usar a coluna B para formatação condicional de A.

Vamos colocar nas columnas C e D os números de linhas, cujas elementos vamos comparar, juntando os blocos ordenados.

	A	B	C	D	E
1		bloco			
2	533	1			
3	252	2			
4	537	2			
5	480	3			
6	576	3			
7	746	3			
8	409	4			
9	857	4			
10	146	5			
11	535	5			
12	52	6			
13	450	6			
14	2	7			
15	426	7			
16	68	8			
17	275	8			
18	612	8			
19	193	9			
20	877	9			

Formatação condicional

☒ 1ª Condição

A fórmula é

Estilo da célula

☐ 2ª Condição

O valor da célula

Estilo da célula

☐ 3ª Condição

O valor da célula

Estilo da célula

Figure 2:

Vamos explicar pelo exemplo. Seja na columna A temos blocos ordenados {63, 193, 438} e {241, 372}. Para juntar esses blocos devemos fazer as conperações seguintes:

	A	B	C	D	I	J
1		bloco	Primeiro	Segundo		
2	63	1	2	5	63	1
3	193	1	3	5	193	1
4	438	1	4	5	241	1
5	241	2	4	6	372	1
6	372	2	4	-1	438	1
7	237	3	7	9	237	2
8	942	3	8	9	521	2
9	521	4	8	10	574	2
10	574	4	8	-1	942	2
11	456	5	11	13	456	3
12	829	5	12	13	588	3

Figure 3:

$$\begin{array}{llll}
\{63, 193, 438\} & \{241, 372\} & 63 * 241 & \longrightarrow 63; \\
\{193, 438\} & \{241, 372\} & 193 * 241 & \longrightarrow 193 \\
\{438\} & \{241, 372\} & 438 * 241 & \longrightarrow 241; \\
\{438\} & \{372\} & 438 * 372 & \longrightarrow 372; \\
\{438\} & \{\emptyset\} & 438 * \{\emptyset\} & \longrightarrow 438;
\end{array}$$

For os valores de conjuntos $\{63, 193, 438\}$ e $\{241, 372\}$ são armazenados nos campos $\{A2, A3, A4\}$ e $\{A5, A6\}$, podemos reescrever a sequência de operações de ordenação na maneira seguinte:

$$\begin{array}{llll}
\{A2 = 63, A3 = 193, A4 = 438\}; & \{A5 = 241, A6 = 372\} & & \\
\{A2, A3, A4\} & \{A5, A6\} & A2 * A5 & \longrightarrow A2; \\
\{A3, A4\} & \{A5, A6\} & A3 * A5 & \longrightarrow A3 \\
\{A4\} & \{A5, A6\} & A4 * A5 & \longrightarrow A5; \\
\{A4\} & \{A6\} & A4 * A6 & \longrightarrow 372; \\
\{A4\} & \{\emptyset\} & A4 * \{\emptyset\} & \longrightarrow A4;
\end{array}$$

Retângulo lagange de Fig. 3 descreve o processo de ordenação de pare mencionada de duas sequências ordenadas. Por exemplo, linha 4 corresponde a comparação

$$\{A4 * A5 \longrightarrow A5 = 241\}.$$

Assim, nossa tarefa de junção de conjuntos $\{63, 193, 438\}$ e $\{241, 372\}$ precisa determinação de elementos de columnas C e D. Lembramos, que os conjuntos $\{63, 193, 438\}$ e $\{241, 372\}$ serão de exemplos, mas tamanho deles não é dado, e deve ser computado pelas formulas com columns A única entrada.

2 Resolução

Começo de proximo bloco ordenado. Pelo como contruirmos planilha, o primeiro bloco ordenado começa em A2.

para essa dinalidade usamos a função **CORRESP**.

CORRESP(valor_procurado, matriz_procurada, [tipo_correspondencia]).

Tipo_correspondencia=0 localiza o primeiro valor que e exatamente igual a valor_procurado.

Com columna B, podemos procurar o começo de proximo bloco como

$$=CORRESP(B2+1;B\$2:B\$31;0)$$

lembrando que os valores para ordenar ficam em A2:A31, e os números de blocos ordenados estão em B2:B31.

Assim, preenchemos as células C2 e D2 com:

$$C2: =LINHA(B2)$$

$$D2: =1+CORRESP(B2+1;B\$2:B\$31;0)$$

Recuperar valor ordenado pelo número de columna. Para esse finalidade vamos usar composição de funções INDIRETO e CONCATENAR.

Função INDIRETO devolve o valor de célula pelo seu argumento. Por exemplo, INDIRETO("A5") devolva o valor (texto ou numero) de célula A5.

Função CONCATENAR concatena seus argumentos. Por exemplo, se H10 contem o número 15, então

$$=INDIRETO(CONCATENAR("A"; H10))$$

vai devolver o conteúdo de A15. Assim, como na Figura 3, devemos comparar os valores de A, com índices de C2 e D2, a expressão correspondente é

$$=INDIRETO(CONCATENAR("A"; C2))<INDIRETO(CONCATENAR("A"; D2))$$

Quando juntamos dois conjuntos ordenados, vai acontecer um momento, quando um desses conjuntos vai acabar. Nesse caso escrevemos no correspondente valor "-1" de columna C ou D. Assim, vamos comparar os valores com números em C10 e D10 como

$$=SE(C10=-1; D10;$$

$$SE(D10=-1; C10;$$

$$SE(INDIRETO(CONCATENAR("A"; C10))<=$$

$$INDIRETO(CONCATENAR("A"; D10));$$

$$C10; D10)))$$

Vamos colocar essa comparação na columna E.

Vamos colocar nas columnas F e G os indicadores de fato, que não há o proximo elemento de primeiro ou segundo conjunto respectivamente.

Essas formulas, para linha 10, serão as seguintes:

$$F10:$$

$$=SE(C10=-1; 0=0;$$

$$INDIRETO(CONCATENAR("B";C10))<>$$

INDIRETO(CONCATENAR("B";C10+1)))

G10:

=SE(D10=-1;0=0;

INDIRETO(CONCATENAR("B";D10))<>

INDIRETO(CONCATENAR("B";D10+1)))




Figure 4:

Proxima pare de elementos para comparar.

O fluxograma de algoritmo de constuição de valores de $C[i+1]$ e $D[i+1]$ pelas C_i e D_i está dada na Fig. 4. Com essa fluxigrama podemos construir as formulas para columnas C e D.

Vamos usar uma columna auxiliar H para o proximo valor (linha para columna D) no caso, se temos um começo de novo conjunto. Essa expressão será similar à aquela de celula D2, mas com uma pequena complicação. Pode acontecer, que o conunto em columna A é últmo e, assim, não há o número de conjunto proximo. Nesse caso a função **CORRESP** vai devolver um erro #N/A. Devemos prender esse erro, e devolver -1 no caso. Assim, temos uma formaula para

celula H10:

H10:

```
=SE(ÉERROS(1+CORRESP(B10+1;B$2:B$31;0));  
-1;  
1+CORRESP(B10+1;B$2:B$31;0))
```

Vamos usar columna H e diagrama de Fig. 4 para formulas de columnas C e D.

C10:

```
=SE(C9=E9;  
SE(F9;  
SE(D9<>-1; -1; LINHA(B10));  
C9+1);  
SE(C9<>-1; C9; SE(G9;LINHA(B10);-1)))
```

D10:

```
=SE(D9=E9;  
SE(G9;  
SE(C9<>-1; -1; SE(F9;H10;-1));  
D9+1);  
SE(D9<>-1; D9; SE(F9;H10;D9)))
```

Assim, constiuirmos mais a columna I pela columna A, cuntando pares dos blocos subse-quentes ordenados de A, e diminuindo a quantidade de blocos ordenados aproximadamente duas vezes (veja Fig. 5).

Vamos **melhorar as formulas desenvolvidas**. Lembramos, que quando usamos a com-posição “**INDIRETO(CONCATENAR...**”, usamos o nome de columna (A ou B) explicito. Isto atrapnlha a possibilidade de copiar essas formulas. A função **INDIRETO** aceita o argu-mento também na maneira RiCj, de acrescentas o segundo algumento “FALSE”. Por exemplo,

=INDIRETO(“R1C2”;1=0)

devolva o valor “bloco”, que está na celula B2, cuja linha (“R” é de palavra ingles “row”)é 1, e columna é 2 (“R” é de palavra ingles “column”).

Assim, usando esse jeuito de função INDIRETO, podemos re-escrever as formulas de colum-nas E, F, G, H, I na maneira seguinte:

E10:

```
=SE(
```

	A	B	C	D	E	F	G	H	I	J
1		bloco	Prim.	Seg.	Prim. /Seg.	Termin. Prim	Termin. Seg.	Prox. Seg.		
2	63	1	2	5	2	FALSE	FALSE	5	63	1
3	193	1	3	5	3	FALSE	FALSE	5	193	1
4	438	1	4	5	5	TRUE	FALSE	5	241	1
5	241	2	4	6	6	TRUE	TRUE	7	372	1
6	372	2	4	-1	4	TRUE	TRUE	7	438	1
7	237	3	7	9	7	FALSE	FALSE	9	237	2
8	942	3	8	9	9	TRUE	FALSE	9	521	2
9	521	4	8	10	10	TRUE	TRUE	11	574	2
10	574	4	8	-1	8	TRUE	TRUE	11	942	2
11	456	5	11	13	11	FALSE	FALSE	13	456	3
12	829	5	12	13	13	TRUE	FALSE	13	588	3
13	588	6	12	14	14	TRUE	TRUE	15	659	3
14	659	6	12	-1	12	TRUE	TRUE	15	829	3
15	274	7	15	17	17	FALSE	FALSE	17	90	4
16	732	7	15	18	15	FALSE	FALSE	17	274	4
17	90	8	16	18	18	TRUE	FALSE	20	290	4
18	290	8	16	19	16	TRUE	TRUE	20	732	4
19	960	8	-1	19	19	TRUE	TRUE	20	960	4
20	575	9	20	23	23	FALSE	FALSE	23	533	5
21	770	9	20	24	20	FALSE	FALSE	23	575	5
22	773	9	21	24	21	FALSE	FALSE	23	770	5
23	533	10	22	24	22	TRUE	FALSE	26	773	5
24	972	10	-1	24	24	TRUE	FALSE	26	972	5
25	998	10	-1	25	25	TRUE	TRUE	26	998	5
26	357	11	26	27	27	TRUE	FALSE	27	184	6
27	184	12	26	28	26	TRUE	TRUE	29	357	6
28	372	12	-1	28	28	TRUE	TRUE	29	372	6
29	219	13	29	30	30	TRUE	FALSE	30	62	7
30	62	14	29	31	29	TRUE	TRUE	-1	219	7
31	354	14	-1	31	31	TRUE	TRUE	-1	354	7

Figure 5:

C10=-1;

D10;

SE(D10=-1;

C10;

SE(INDIRETO(CONCATENAR("R";C10;"C";COLUNA(A10));0)<=
INDIRETO(CONCATENAR("R";D10;"C"; COLUNA(A10));0);
C10;D10)))

F10:

=SE(

C10=-1;0=0;

INDIRETO(CONCATENAR("R";C10;"C";COLUNA(B10));0) <>

INDIRETO(CONCATENAR("R";C10+1;"C";COLUNA(B10));0))

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R
1		bloco	I	II		bloco	I	II		bloco	I	II		bloco	I	II		bloco
2	63	1	2	5	63	1	2	7	63	1	2	11	63	1	2	20	62	1
3	193	1	3	5	193	1	3	7	193	1	3	11	90	1	2	21	63	1
4	438	1	4	5	241	1	4	7	237	1	3	12	193	1	3	21	90	1
5	241	2	4	6	372	1	4	8	241	1	4	12	237	1	4	21	184	1
6	372	2	4	-1	438	1	5	8	372	1	5	12	241	1	4	22	193	1
7	237	3	7	9	237	2	6	8	438	1	6	12	274	1	5	22	219	1
8	942	3	8	9	521	2	-1	8	521	1	6	13	290	1	5	23	237	1
9	521	4	8	10	574	2	-1	9	574	1	6	14	372	1	6	23	241	1
10	574	4	8	-1	942	2	-1	10	942	1	7	14	438	1	7	23	274	1
11	456	5	11	13	456	3	11	15	90	2	8	14	456	1	8	23	290	1
12	829	5	12	13	588	3	11	16	274	2	8	15	521	1	9	23	354	1
13	588	6	12	14	659	3	11	17	290	2	9	15	574	1	9	24	357	1
14	659	6	12	-1	829	3	11	18	456	2	10	15	588	1	9	25	372	1
15	274	7	15	17	90	4	12	18	588	2	10	16	659	1	10	25	372	1
16	732	7	15	18	274	4	13	18	659	2	10	17	732	1	10	26	438	1
17	90	8	16	18	290	4	14	18	732	2	10	18	829	1	11	26	456	1
18	290	8	16	19	732	4	14	19	829	2	10	19	942	1	12	26	521	1
19	960	8	-1	19	960	4	-1	19	960	2	-1	19	960	1	13	26	533	1
20	575	9	20	23	533	5	20	26	184	3	20	29	62	2	13	27	574	1
21	770	9	20	24	575	5	20	27	357	3	20	30	184	2	14	27	575	1
22	773	9	21	24	770	5	20	28	372	3	21	30	219	2	14	28	588	1
23	533	10	22	24	773	5	20	-1	533	3	21	31	354	2	15	28	659	1
24	972	10	-1	24	972	5	21	-1	575	3	21	-1	357	2	16	28	732	1
25	998	10	-1	25	998	5	22	-1	770	3	22	-1	372	2	17	28	770	1
26	357	11	26	27	184	6	23	-1	773	3	23	-1	533	2	17	29	773	1
27	184	12	26	28	357	6	24	-1	972	3	24	-1	575	2	17	30	829	1
28	372	12	-1	28	372	6	25	-1	998	3	25	-1	770	2	18	30	942	1
29	219	13	29	30	62	7	29	-1	62	4	26	-1	773	2	19	30	960	1
30	62	14	29	31	219	7	30	-1	219	4	27	-1	972	2	-1	30	972	1
31	354	14	-1	31	354	7	31	-1	354	4	28	-1	998	2	-1	31	998	1

Figure 6:

G10:

=SE(D10=-1;0=0;

INDIRETO(CONCATENAR("R";D10;"C";COLUNA(B10));0) <>

INDIRETO(CONCATENAR("R";D10+1;"C";COLUNA(B10));0))

H10:

=SE(

ÉERROS(1 +CORRESP(B10+1;B\$2:B\$31;0));

-1;

1 +CORRESP(B10+1;B\$2:B\$31;0))

I10:

=INDIRETO(CONCATENAR("R";E10;"C";COLUNA(A10));0)

Podemos **eliminar as columnas auxiliaadoras** E, F, G, e H. Isto significa, que vamos copiar as formulas dessas columnas para cada lugar, onde essas formulas são mencionadas. Depois dessa eliminação obtemos as formulas seguintes:

C2:

LINHA(B2)

C3:

SE(C2=SE(C2=-1;D2;SE(D2=-1;C2;SE(INDIRETO(CONCATENAR("R";C2;"C";COLUNA(A2));0)<=INDIRETO(CONCATENAR("R";D2;"C";COLUNA(A2));0);C2;D2));SE(SE(C2=-1;0=0;INDIRETO(CONCATENAR("R";C2;"C";COLUNA(B2));0)<>INDIRETO(CONCATENAR("R";C2+1;"C";COLUNA(B2));0));SE(D2<>-1;-1;LINHA(B3));C2+1);SE(C2<>-1;C2;SE(SE(D2=-1;0=0;INDIRETO(CONCATENAR("R";D2;"C";COLUNA(B2));0)<>INDIRETO(CONCATENAR("R";D2+1;"C";COLUNA(B2));0));LINHA(B3);-1)))

Deve copiar C3 até fim da lista.

D2:

1+CORRESP(B2+1;B\$2:B\$31;0)

D3:

SE(D2=SE(C2=-1;D2;SE(D2=-1;C2;SE(INDIRETO(CONCATENAR("R";C2;"C";COLUNA(A2));0)<=INDIRETO(CONCATENAR("R";D2;"C";COLUNA(A2));0);C2;D2));SE(SE(D2=-1;0=0;INDIRETO(CONCATENAR("R";D2;"C";COLUNA(B2));0)<>INDIRETO(CONCATENAR("R";D2+1;"C";COLUNA(B2));0));SE(C2<>-1;-1;SE(SE(C2=-1;0=0;INDIRETO(CONCATENAR("R";C2;"C";COLUNA(B2));0)<>INDIRETO(CONCATENAR("R";C2+1;"C";COLUNA(B2));0));SE(ÉERROS(1+CORRESP(B3+1;B\$2:B\$31;0));-1;1+CORRESP(B3+1;B\$2:B\$31;0));-1));D2+1);SE(D2<>-1;D2;SE(SE(C2=-1;0=0;INDIRETO(CONCATENAR("R";C2;"C";COLUNA(B2));0)<>INDIRETO(CONCATENAR("R";C2+1;"C";COLUNA(B2));0));SE(ÉERROS(1+CORRESP(B3+1;B\$2:B\$31;0));-1;1+CORRESP(B3+1;B\$2:B\$31;0));D2)))

Deve copiar D3 até fim da lista.

E2:

```
INDIRETO(CONCATENAR("R";SE(C2=-1;D2;SE(D2=-1;C2;  
SE(INDIRETO(CONCATENAR("R";C2;"C";COLUNA(A2));0) <=  
INDIRETO(CONCATENAR("R";D2;"C";COLUNA(A2));0);C2;D2)));  
"C";COLUNA(A2));0)
```

Deve copiar E2 até fim da lista.

Depois de todas essas operações podemos copiar as computações para direiro, aplicando-as para columnas obtidas na cada passo, até finalizar ordenação. O resultado dessas aplicações está na Fig. 6.